Ćwiczenia (z przykładami i częściowymi rozwiązaniami)

1.

Narysuj wykresy funkcji $z = x^2 + y^2$, $z = 2x + 2y - 9$ w R^3. Narysuj linie $x = 1 + 2t$, $y = 1 + 2t$, $z = 2 - t$. Pokaż wszystkie wykresy na jednym obrazku.

Rozwiązanie

Plot3D[{x^2 + y^2, 2 x + 2 y - 9}, {x, -10, 10}, {y, -10, 10}]



ParametricPlot3D[{1 + 2 t, 1 + 2 t, 2 - t}, {t, -10, 10}]



```
Show[%, %%]
```



2.

Sprawdź następujące twierdzenie: niech det A będzie wyznacznikiem $n \times n$ macierzy A (n ustalone). Oznaczmy przez $\det A(j, k)$ wyznacznik macierzy otrzymanej przez usunięcie j-tego szeregu i k-tej kolumny z macierzy A. Podobnie, możemy usunąć dalsze szeregi i kolumny. Sprawdź identyczność $\det A \det A(i, j, k, l) = \det\left[\begin{pmatrix} \det A(i, j) & \det A(i, l) \\ \det A(k, j) & \det A(k, l) \end{pmatrix}\right]$. Zobacz poniższy przykład dla $n = 5$, $[i, j, k, l] = [1, 2, 4, 5]$. Zilustruj to twierdzenie przy pomocy Manipulate dając użytnikowi wybór którą kolumnę lub szereg usunąć.

Przykład

```
Table[a[i, j], {i, 1, 5}, {j, 1, 5}] // MatrixForm
```

$$\begin{pmatrix} a[1,1] & a[1,2] & a[1,3] & a[1,4] & a[1,5] \\ a[2,1] & a[2,2] & a[2,3] & a[2,4] & a[2,5] \\ a[3,1] & a[3,2] & a[3,3] & a[3,4] & a[3,5] \\ a[4,1] & a[4,2] & a[4,3] & a[4,4] & a[4,5] \\ a[5,1] & a[5,2] & a[5,3] & a[5,4] & a[5,5] \end{pmatrix}$$

$$\mathbf{A = Det}\left[\begin{pmatrix} a[1,1] & a[1,2] & a[1,3] & a[1,4] & a[1,5] \\ a[2,1] & a[2,2] & a[2,3] & a[2,4] & a[2,5] \\ a[3,1] & a[3,2] & a[3,3] & a[3,4] & a[3,5] \\ a[4,1] & a[4,2] & a[4,3] & a[4,4] & a[4,5] \\ a[5,1] & a[5,2] & a[5,3] & a[5,4] & a[5,5] \end{pmatrix}\right]$$

$$\mathbf{A45 = Det}\left[\begin{pmatrix} a[1,1] & a[1,2] & a[1,3] & a[1,4] \\ a[2,1] & a[2,2] & a[2,3] & a[2,4] \\ a[3,1] & a[3,2] & a[3,3] & a[3,4] \\ a[5,1] & a[5,2] & a[5,3] & a[5,4] \end{pmatrix}\right]$$

$$\mathbf{A12 = Det}\left[\begin{pmatrix} a[2,1] & a[2,3] & a[2,4] & a[2,5] \\ a[3,1] & a[3,3] & a[3,4] & a[3,5] \\ a[4,1] & a[4,3] & a[4,4] & a[4,5] \\ a[5,1] & a[5,3] & a[5,4] & a[5,5] \end{pmatrix}\right]$$

```
A1245 = Det[(a[2, 1]  a[2, 3]  a[2, 4]
             a[3, 1]  a[3, 3]  a[3, 4]
             a[5, 1]  a[5, 3]  a[5, 4])]
```

$$A1245 = \mathrm{Det}\left[\begin{pmatrix} a[2,1] & a[2,3] & a[2,4] \\ a[3,1] & a[3,3] & a[3,4] \\ a[5,1] & a[5,3] & a[5,4] \end{pmatrix}\right]$$

$$A15 = \mathrm{Det}\left[\begin{pmatrix} a[2,1] & a[2,2] & a[2,3] & a[2,4] \\ a[3,1] & a[3,2] & a[3,3] & a[3,4] \\ a[4,1] & a[4,2] & a[4,3] & a[4,4] \\ a[5,1] & a[5,2] & a[5,3] & a[5,4] \end{pmatrix}\right]$$

$$A42 = \mathrm{Det}\left[\begin{pmatrix} a[1,1] & a[1,3] & a[1,4] & a[1,5] \\ a[2,1] & a[2,3] & a[2,4] & a[2,5] \\ a[3,1] & a[3,3] & a[3,4] & a[3,5] \\ a[5,1] & a[5,3] & a[5,4] & a[5,5] \end{pmatrix}\right]$$

```
A A1245 - Det[{{A12, A15}, {A42, A45}}] // Together
0
```

3.

Przeczytaj o twierdzeniu Sturma (http://en.wikipedia.org/wiki/Sturm%27s_theorem) i implementuj algorytm który znajduje ilość różnych rzeczywistych pierwiastków wielomianu bez podwójnych pierwiastków.

Przykład

Następujący wielomian ma dwa rzeczywiste pierwiastki

$x^4 + x^3 - x - 1$ // Factor

$(-1 + x)\ (1 + x)\ (1 + x + x^2)$

Typowe podejście do programowania funkcyjnego składa się z implimentacji algorytmu krok po kroku poczym połączeniu pojedyńczych kroków w jedną funkcję

Definiujemy wielomian

q[x_] := $x^4 + x^3 - x - 1$

Sprawdzamy że wielomian nie ma podwójnych pierwiastków:

```
Discriminant[q[x], x] ≠ 0
True
```

```
p[0] = q[x]
```

$-1 - x + x^3 + x^4$

```
p[1] = D[p[0], x]
```

$-1 + 3\,x^2 + 4\,x^3$

```
PolynomialRemainder[p[0], p[1], x]
```

$$-\frac{15}{16} - \frac{3\,x}{4} - \frac{3\,x^2}{16}$$

```
p[2] = -PolynomialRemainder[p[0], p[1], x] // Simplify
```

$$\frac{3}{16}\left(5+4x+x^2\right)$$

```
p[3] = -PolynomialRemainder[p[1], p[2], x] // Simplify
```

$$-32\,(2+x)$$

```
p[4] = -PolynomialRemainder[p[2], p[3], x] // Simplify
```

$$-\frac{3}{16}$$

```
p[5] = -PolynomialRemainder[p[3], p[4], x] // Simplify
```

0

Łączymy wszystko w jedną funkcję. Tu wygodnie jest użyć funkcji NestWhileList:

```
? NestWhileList
```

NestWhileList[*f*, *expr*, *test*] generates a list of the results of applying *f* repeatedly, starting with *expr*, and continuing until applying *test* to the result no longer yields True.
NestWhileList[*f*, *expr*, *test*, *m*] supplies the most recent *m* results as arguments for *test* at each step.
NestWhileList[*f*, *expr*, *test*, All] supplies all results so far as arguments for *test* at each step.
NestWhileList[*f*, *expr*, *test*, *m*, *max*] applies *f* at most *max* times. ≫

```
SturmSequence[f_, x_] /; PolynomialQ[f, x] && Discriminant[f, x] ≠ 0 :=
 Prepend[NestWhileList[{#[[2]], -PolynomialRemainder[#[[1]], #[[2]], x]} &,
    {f, D[f, x]}, ! FreeQ[Last[#], x] &][[All, 2]], f]
```

```
ss = SturmSequence[q[x], x]
```

$$\left\{-1-x+x^3+x^4,\ -1+3x^2+4x^3,\ \frac{15}{16}+\frac{3x}{4}+\frac{3x^2}{16},\ -64-32x,\ -\frac{3}{16}\right\}$$

Czołowe współczynniki łańcucha Sturma:

```
CoefficientList[#, x][[-1]] & /@ ss // FullSimplify
```

$$\left\{1,\ 4,\ \frac{3}{16},\ -32,\ -\frac{3}{16}\right\}$$

To samo po zastąpineniu $x \to -x$:

```
CoefficientList[#, x][[-1]] & /@ (ss /. x → -x) // FullSimplify
```

$$\left\{1,\ -4,\ \frac{3}{16},\ 32,\ -\frac{3}{16}\right\}$$

Inne pożyteczne funkcje

Sprawdzamy dla jakich parametrów *a* wielomian nie ma pierwiastków rzeczywistych.

```
Resolve[ForAll[x, Element[x, Reals], a - 8 x + a x^3 + x^4 != 0], Reals]
```

False

Sprawdzamy dla jakich parametrów *a* wielomian nie ma pierwiastków rzeczywistych.

```
Resolve[ForAll[x, Element[x, Reals], a - 8 x + a x^3 + x^4 >= 0], Reals]
```

a == 4

Czyli dla $a = 4$ wielomian jest nie ujemny i ma pierwiaski rzeczywiste. Sprawdźmy ile ich ma:

```
CountRoots[a - 8 x + a x^3 + x^4 /. a -> 4, x]
```

4

```
Plot[a - 8 x + a x^3 + x^4 /. a -> 4, {x, -3, 3}]
```



4.

1. Niech Q[x] będzie wielomianem stopnia
  n. Znajdź wartości charakterystyczne macierzy

$$\begin{pmatrix} s(0) & s(1) & s(2) & s(3) & \ldots & s(n) \\ s(1) & s(2) & s(3) & s(4) & \ldots & s(n+1) \\ s(2) & s(3) & s(4) & s(5) & \ldots & s(n+2) \\ s(3) & s(4) & s(5) & s(6) & \ldots & s(n+3) \\ \ldots & \ldots & \ldots & \ldots & \ldots & \ldots \\ s(n) & s(n+1) & s(n+2) & s(n+3) & \ldots & s(2n) \end{pmatrix}$$

o wymiarach $n+1 \times n+1$ gdzie $s(k)$ są śladami iloczynów macierzy $C_1{}^k = C_1 \cdots C_1$ i $s(0) = n$. Macierz C_1 o wymiarach $n \times n$ jest dana przez dołączenie współczynników wielomianu znakiem minus do ostatniego szeregu a reszta macierzy jest określona przez warunek że jedyne nie zerowe elementy są rowne 1 i znajdują się w pozycjach $(i, i+1)$.

2. Znajdź sygnaturę macierzy S (liczbę dodatnich i ujemnych wartości własnych). Zauważ że ponieważ macierz jest symnetryczna, jej wartości własne są rzeczywiste.

Przykład.

```
Clear[q, s, ss]
```

```
q[x_] := 2 - 8 x + 3 x^3 + x^4
```

```
l1 = Insert[#, 0, 1] & /@ IdentityMatrix[3]
```

```
{{0, 1, 0, 0}, {0, 0, 1, 0}, {0, 0, 0, 1}}
```

```
l2 = Most[-CoefficientList[Q[x], x]]
```

{-2, 8, 0, -3}

```
(C1 = Append[l1, l2]) // TraditionalForm
```

$$\begin{pmatrix} 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \\ -2 & 8 & 0 & -3 \end{pmatrix}$$

```
s[1] = Tr[C1]
```

-3

```
ss[1] = C1
```

{{0, 1, 0, 0}, {0, 0, 1, 0}, {0, 0, 0, 1}, {-2, 8, 0, -3}}

```
ss[k_?(# > 1 &)] := C1.ss[k - 1]
s[0] = 4;
s[k_?(# > 0 &)] := Tr[ss[k]]
ss[2]
```

{{0, 0, 1, 0}, {0, 0, 0, 1}, {-2, 8, 0, -3}, {6, -26, 8, 9}}

```
s[2] // Simplify
```

9

```
(mm = Map[s, Table[Range[i, i + 3], {i, 0, 3}], {2}]) // TraditionalForm
```

$$\begin{pmatrix} 4 & -3 & 9 & -3 \\ -3 & 9 & -3 & -23 \\ 9 & -3 & -23 & 147 \\ -3 & -23 & 147 & -483 \end{pmatrix}$$

```
Eigenvalues[mm] // N
```

{-526.862, 28.2535, 4.9197, 0.689249}

5. Równania różniczkowe zwyczajne [1].

1.

Narysuj portret fazowy

Rozwiązanie.

```
v = NDSolve[{x'[t] == x[t] + y[t], y'[t] == -x[t], x[0] == 1, y[0] == 0},
  {x[t], y[t]}, {t, 0, 10}]
```

{{x[t] → InterpolatingFunction[{{0., 10.}}, <>][t],
  y[t] → InterpolatingFunction[{{0., 10.}}, <>][t]}}

```
ParametricPlot[Evaluate[{x[t], y[t]} /. v],
 {t, 0, 10}, PlotRange → All, PlotPoints → 1000]
```



2.

Narysuj pole wektorowe i krzywe fazowe dla układu RRZ 2 – go rzędu

Solution.

```
p1 = VectorPlot[{y, -x}, {x, -2, 2}, {y, -2, 2}, PlotRange → All]
```




```
ode[x0_, y0_] := NDSolve[
  {x'[t] == y[t], y'[t] == -x[t], x[0] == x0, y[0] == y0}, {x[t], y[t]}, {t, -200, 200}]

sol[i_] := ode[0.22 i, 0.2 i]

p2 = ParametricPlot[Evaluate[Table[{x[t], y[t]} /. sol[i], {i, 1, 6}]],
  {t, 0, 200}, AxesLabel → {"x", "y"}]
```



```
Show[{p1, p2}]
```



6. Geometria różniczkowa [2].

1.

Obliczyć parametr łukowy krzywej w R^3.

Rozwiązanie.

```
arclength[α_List, t_] := Integrate[Sqrt[Simplify[D[α, t].D[α, t]]], t]
```

```
arclength[{2 Cos[t], 2 Sin[t], 1}, t]
```

```
2 t
```

2.

Narysuj kilka stycznych do okręgu

Rozwiązanie.

```
tangentline[α_List, s_, t_] :=
 Line[{α, α + s D[α, t] / Sqrt[Factor[D[α, t].D[α, t]]]}]
```

```
tt = tangentline[{Cos[t], Sin[t]}, 1, t]
```

$$\text{Line}\left[\left\{\{\text{Cos}[t], \text{Sin}[t]\}, \left\{\text{Cos}[t] - \text{Sin}[t] \Big/ \left(\sqrt{\left(\text{Cos}[t]^2 + \text{Sin}[t]^2\right)}\right), \text{Sin}[t] + \text{Cos}[t] \Big/ \left(\sqrt{\left(\text{Cos}[t]^2 + \text{Sin}[t]^2\right)}\right)\right\}\right\}\right]$$

```
Show[Graphics[{AbsoluteThickness[1], Evaluate[Table[tt, {t, 0, 2 π, π / 36}]]}]]
```



3.

Narysuj krzywą płaską sparametryzowaną parametrem łukowym z zadaną krzywizną κ.
Wskazówka: rozwiąż układ RRZ x'[s] == Cos[θ[s]], y'[s] == Sin[θ[s]], θ'[s] == κ.

Rozwiązanie.

```
curve[κ_, s_, a_: 0, {c_: 0, d_: 0, theta0_: 0}, {smin_: -10, smax_: 10}] :=
 Flatten[{x[s], y[s]} /. NDSolve[{x'[s] == Cos[θ[s]], y'[s] == Sin[θ[s]], θ'[s] == κ,
     x[a] == c, y[a] == d, θ[a] == theta0}, {x[s], y[s], θ[s]}, {s, smin, smax}]]
```

```
curve[(s + Sin[s]), s, 0, {0, 0, 0}, {-18, 18}]
```

```
{InterpolatingFunction[{{-18., 18.}}, <>][s],
 InterpolatingFunction[{{-18., 18.}}, <>][s]}
```

```
ParametricPlot[Evaluate[%], {s, -18, 18}, AspectRatio → Automatic]
```



4.

Obliczyć krzywiznę średnią oraz Gaussa krzywej w R^3.

7.

Narysuj obrazy rzutu stereograficznego punktów z R^2 (lub płaszczyzny zespołonej C) na sferze Riemanna (http://en.wikipedia.org/wiki/Stereographic_projection).

8.

Zaimplementuj metodę Frobeniusa dla RRZ (http://en.wikipedia.org/wiki/Frobenius_method, http://-mathworld.wolfram.com/FrobeniusMethod.html).

9.

Zaimplemenuj hipotezę ABC (http://en.wikipedia.org/wiki/Abc_conjecture).

Literatura.

[1] S. Lynch, Dynamical Systems with Applications using Mathematica®, Birkhäuser, Basel, 2007, XV, 484 p.

[2] A. Gray, Modern Differential Geometry of Curves and Surfaces with *Mathematica*, CRC Press

LLC, 1998. Pliki *Mathematica* można pobrać ze strony http://library.wolfram.com/infocenter/Books/3759.